
FINISH 
/CLEAR 
C***************************************************************************

********* 
C*** 
C*** RUOTA ANALIZZATA CON MODELLO 3D 
C*** 
C*** PARAMETRI 
C*** 
/PREP7 
DA=90          ! diametro albero 
DM=130          ! diametro mozzo 
DB=350           ! diametro anello esterno 
DR=400           ! daiametro ruota 
LA= 200          ! lunghezza albero 
LM=130          ! lunghezza mozzo 
LR=120           ! lunghezza anello esterno 
RR=15             ! raggio raccordo 
SC=50             ! spessore cartella 
FR=20000           ! carico 
C*** 
C*** KEYPONTS 
C*** 
K,1, 
K,2,DA/2 
K,3,DA/2,LA/2 
K,4,0,LA/2 
K,5,DR/2 
K,6,DR/2,LR/2 
K,7,DB/2,LR/2 
K,8,DB/2,SC/2 
K,9,DM/2,SC/2 
K,10,DM/2,LM/2 
K,11,DA/2,LM/2 
C*** 
C*** LINEE 
C*** 
L,1,2 
L,2,11 
L,11,3 
L,3,4 
L,4,1 
L,2,5 
L,5,6 
L,6,7 
L,7,8 
L,8,9 
L,9,10 
L,10,11 
C*** 
C*** RACCORDI 
C*** 
LSEL,,LOC,X,DM/2+0.1,DB/2-0.1 
LSEL,R,LOC,Y,0.1,SC/2+0.1 
*GET,NL1,LINE,,NUM,MAX 
LSEL,,LOC,X,DB/2-0.1,DB/2+0.1 
LSEL,R,LOC,Y,SC/2-0.1,LR/2+0.1 
*GET,NL2,LINE,,NUM,MAX 
LSEL,ALL 
LFILLT,NL1,NL2,RR 
LSEL,,LOC,X,DM/2-0.1,DM/2+0.1 
LSEL,R,LOC,Y,SC/2-0.1,LM/2+0.1 
*GET,NL2,LINE,,NUM,MAX 



LSEL,ALL 
LFILLT,NL1,NL2,RR 
C*** 
C*** AREE 
C*** 
LSEL,,LOC,X,DA/2-0.1,99999 
LSEL,R,LOC,Y,-0.1,LM/2+0.01 
AL,ALL 
LSEL,,LOC,X,-0.1,DA/2+0.01 
AL,ALL 
C*** 
C*** ELEMENTI 
C*** 
ET,1,25 
SMRTSIZE,3 
AMESH,ALL 
ET,2,45 
TYPE,2 
C*** ESIZE,,12 
C*** 
C*** PROPRIETA' MATERIALE 
C*** 
MP,EX,1,210000   
C*** 
C*** VOLUMI 
C*** 
EXTOPT,ON 
EXTOPT,ESIZE,18,4 
VROTAT,1,2,,,,,1,4,90 
ASEL,,LOC,X,-0.001,0.001 
EXTOPT,ESIZE,6,2 
VROTAT,ALL,,,,,,1,4,90 
ASEL,ALL 
C*** 
C*** VINCOLI 
C*** 
ASEL,,LOC,Y,-0.1,0.01         ! vincoli di simmetria sul piano medio 
DA,ALL,SYMM 
ASEL,,LOC,Z,-0.1,0.01         ! vincoli di simmetria sul piano assiale 
DA,ALL,SYMM 
ASEL,ALL 
NSEL,,LOC,Z,-1,0.001          ! vincoli sull'asse 
NSEL,R,LOC,X,-0.001,0.001 
D,ALL,UX,0 
NSEL,ALL 
C*** 
C*** CARICHI 
C*** 
NSEL,,LOC,Z,-1,0.001                          ! seleziona i nodi  
NSEL,R,LOC,X,DR/2-0.001,DR/2+1 
*GET,NN,NODE,,COUNT                     ! conta i nodi 
F,ALL,FX,-FR/4/(NN-1)/2                     ! carico sui nodi di estremità 
NSEL,R,LOC,Y,0.001,LR/2-0.001 
FDELE,ALL,ALL 
F,ALL,FX,-FR/4/(NN-1)                        ! carico sui nodi di mezzeria 
NSEL,ALL 
FINISH 
 
 
 
 
 


